
Abstract 

Ein Generalisiertes Lineares Modell wird durch den Linearen Prädiktor, die Link-Funktion und die 
Varianz-Funktion definiert. Die Modellauswahl, d.h. die inhaltliche Ausgestaltung dieser drei 
Elemente, kann auf theoretischen Überlegungen oder empirischen Erkenntnissen basieren. Auf 
Fragen der empirischen Modellauswahl mit PROC GENMOD wird im Folgenden eingegangen. 

1. Einleitung 
Die Familie der Generalisierten Linearen Modelle (GLMs) wurde erstmalig von Nelder & 
Wedderburn [1972] zusammenhängend dargestellt. Wie der Name andeutet, stellen GLMs 
eine Erweiterung des klassischen linearen Modells dar. Viele weitere bekannte Modelle, wie 
das Logit-, Probit- oder loglineare Modell, gehören zu der Familie der GLMs. 
 
Mit vielen SAS-Prozeduren (u.a. REG, GLM, LOGISTIC, PROBIT) können einzelne Modelle 
aus der Familie der GLMs modelliert werden. PROC GENMOD eignet sich nicht nur zur 
Modellierung einzelner GLMs, sondern vor allem zur Modellauswahl bzw. zum Modell-
vergleich. Nach einer kurzen Einführung der theoretischen Grundlagen der GLMs wird auf 
einige spezielle Optionen von PROC GENMOD zur Modellauswahl eingegangen. 
 
Was ist ein GLM 
 
Das klassische lineare Modell als der bekannteste Vertreter der Familie der GLMs ist durch 

gegeben. Hierbei bezeichnen yi die Responsevariable bzw. xi den Vektor der 
nichtstochastischen erklärenden Variablen für die i-te der insgesamt T Beobachtungen und 

 den zu schätzenden Parametervektor mit insgesamt P Elementen. Weiterhin wird 
angenommen, dass die unbekannten Störterme ei und damit auch die Responsevariablen yi 
unabhängig normalverteilt sind. 
Die Responsevariable yi kann also in eine systematische Komponente, die durch ihren 
Erwartungswert iµ  gegeben ist, und einen Störterm ei untergliedert werden. Die 

systematische Komponente iµ  wird durch eine (in diesem Fall sehr einfache, d.h. lineare) 
Funktion des linearen Prädiktors xi‘  beschrieben: 

Mit dieser Darstellung sind die wichtigsten Eigenschaften eines GLMs, so wie sie sich im 
klassischen linearen Modell ausprägen, angesprochen. Allgemein wird ein GLM durch 
folgende drei Elemente charakterisiert: 
 
♦ Der lineare Prädiktor η  als Linearkombination der erklärenden Variablen ist analog 

dem klassischen linearen Modell gegeben durch:  
   x ’ii =η  
 

♦ Eine bijektive, zweimal stetig differenzierbare Link-Funktion g beschreibt die Verbindung 
zwischen dem Erwartungswert µ  der Responsevariablen und dem linearen Prädiktor η : 

   )’(.’)( 1 xx iiii gbzwg −== µµ   (1) 

 
♦ Die Dichte- bzw. Wahrscheinlichkeitsfunktion ),;( φθyf  der Responsevariablen y  gehört 

zur Familie der einparametrigen natürlichen Exponentialverteilungen. Die für die Schät-
zung von  wesentliche Eigenschaft der Verteilung ist neben dem Erwartungswert µ  

die Varianzfunktion )(µV , die den Einfluss des Erwartungswerts auf die Varianz der 
Responsevariablen beschreibt:  

Der Dispersionsparameter φ  ist konstant über alle Beobachtungen. Er ist z.B. bei der 
Poisson-Verteilung mit φ  = 1 a-priori bekannt oder muss z.B. im klassischen linearen 
Modell geschätzt werden. Über die Beobachtungsgewichte w kann der Einfluss der 
Beobachtungen auf das Modell individuell bestimmt werden. 

 
Die Familie der Exponentialverteilungen umfasst sowohl stetige Verteilungen, wie die 
Normal- und Gammaverteilung, als auch diskrete Verteilungen, wie die Binomial- und 
Poissonverteilung. Die Wahl einer geeigneten Verteilung ergibt sich oft implizit durch theo-
retische Modellüberlegungen oder die Struktur des Datenmaterials. 
 
Für jede dieser Verteilungen existiert eine sog. kanonische Linkfunktion, die vorteilhafte 
statistische Eigenschaften aufweist und die Modellschätzung vereinfacht. Die Linkfunktion 
sollte auch den unbeschränkten Wertebereich des linearen Prädiktors auf den ggf. 
beschränkten Wertebereich (wie z.B. bei der Binomialverteilung) der Responsevariablen 
abbilden. Zudem sollte die Modellanpassung bei der Auswahl der Link-Funktion berück-
sichtigt werden. 
 

Verteilung V( )µ  KANONISCHE LINKFUNKTION 

Normal 1 η µ=  

Poisson µ  ( )η µ= log  

Gamma µ 2  η µ= −1  

Invers-Normal 
3µ  η µ= −2  

Binomial )1( µµ −  [ ]η µ µ= − −log ( )1 1  

Tabelle 1: Varianzfunktion und kanonische Linkfunktion wichtiger GLM 
 
In Tabelle 1 sind die wichtigsten Verteilungen aus der Familie der Exponentialverteilungen 
mit ihren Varianzfunktionen und kanonischen Linkfunktionen zusammengefasst. 

Devianz und verallgemeinerte X²-Statistik 
 
Jedes GLM definiert eine Log-Likelihood-Funktion ),;( φyl , die in Abhängigkeit von  
maximiert wird. Die nummerische Lösung der Maximum-Likelihood-Schätzung kann mit einer 
iterativen Kleinste-Quadrate-Schätzung, wie z.B. einem Newton-Rhapson-Algorithmus, 
erfolgen. Aus der Schätzung leiten sich mit der Devianz );( yD µ  und der verallgemeinerten 
X²-Statistik zwei für die Modellbeurteilung zentrale Statistiken ab. Die Devianz 
 
 ));();((2);( ylyylyD µφµ −=  (3) 
 
ist eine Funktion der Differenz zwischen der modellunabhängigen maximalen Log-Likelihood 
und der Log-Likelihood des betrachteten Modells. Der ML-Schätzer für den Parametervektor 

 ist gleichzeitig der Devianz-minimierende Schätzer. Jede Varianzfunktion )(µV  definiert 
genau eine Devianz-Funktion. Die verallgemeinerte X²-Statistik 

ist eine quadratische Approximation der Devianz. Aus der verallgemeinerten X²-Statistik wird 
zudem ersichtlich, dass in die Parameterschätzung lediglich die durch (2) gegebene Erwar-
tungswert- und Varianzbeziehung der Exponentialverteilungen eingehen. Weiterhin wird 
deutlich, dass die Varianzfunktion )(µV  das Skalenniveau beider Statistiken bestimmt. 

 

Verteilung );( yD µ  

Normal ( )2
iii yw µ−Σ  

Poisson ( ))()/log(2 iiiiii yyyw µµ −−Σ  

Gamma ( ))/log(/)(2 iiiiii yyw µµµ −−Σ  

Invers-Normal ( ))/()( 22
iiiii yyw µµ−Σ  

Tabelle 2: Devianz-Funktionen 
 
Die skalierte Devianz bzw. skalierte verallgemeinerte X²-Statistik ergibt sich durch Division 
der jeweiligen unskalierten Statistiken durch den Dispersionsparameter: 

Beide skalierte Statistiken folgen unter gewissen Regularitätsannahmen asymptotisch einer 
X²-Verteilung mit I-P Freiheitsgraden. Aus diesen Zusammenhängen leiten sich zwei 
konsistente Schätzer für den Dispersionsparameter φ  ab:  

Im klassischen linearen Modell entspricht die Devianz );( yD µ  und die verallgemeinerte X²-
Statistik der Summe der Abweichungsquadrate (vgl. Tab. 2) und der Dispersionsparameter 
φ  der unbekannten Varianz ²σ  der Responsevariablen. Aus (3) leitet sich auch die Verallge-
meinerung des Likelihood-Ratio-Tests zum Vergleich zweier geschachtelter Modelle ab: 

In diesem Fall ist das Modell bµ  mit b zu schätzenden Parametern eine Vereinfachung des 

Modells aµ  mit a zu schätzenden Parametern. Diese Teststatistik folgt asymptotisch einer 
X²-Verteilung mit a-b Freiheitsgraden. Ist der Dispersionsparameter φ  z.B. durch (5a) zu 
schätzen, folgt die Teststatistik asymptotisch einer F-Verteilung 

und wird als „Verallgemeinerter F-Test“ bezeichnet. Beide Teststatistiken eignen sich sowohl 
zur Bewertung des linearen Prädiktors, z.B. bei einer schrittweisen Variablenselektion, als 
auch zum Test von Parametern in der Link- und Varianzfunktion. 
 
Power-Varianz und Power-Link-Funktionen 
 
Für die Auswahl der optimalen Link- bzw. Varianzfunktion ist es hilfreich, dass alle rele-
vanten Link- und Varianzfunktionen (siehe Tab. 1) zu der Familie der Power-Link-Funktionen 

( )ξµη ;  bzw. Power-Varianz-Funktionen ( )ψµ;V  zählen: 

Wie aus Tabelle 3 zu ersehen ist, umfasst die Familie der Power-Link- bzw. Power-Varianz-
Funktionen alle wichtigen GLMs in ihren kanonischen Formen. 
 

Verteilung ),( ψµV  ),( ξµη  

Normal 0=ψ  1=ξ  

Poisson 1=ψ  0=ξ  

Gamma 2=ψ  1−=ξ  

Invers-Normal 3=ψ  2−=ξ  

Tabelle 3 GLMs mit Power-Link- bzw. Power-Varianz-Funktionen 
 
Andere Kombinationen von ψ  und ξ , die zu keiner Verteilung aus der Exponentialfamilie 
gehören, werden als Quasi-Likelihood-Modelle bezeichnet. Da in die Modellschätzung ledig-
lich die Erwartungswert- und Varianzbeziehungen aus (2) eingehen, gelten für die Parame-
terschätzer alle wesentlichen Eigenschaften der GLM-Schätzer zumindest asymptotisch. 
Diese sog. Quasi-Likelihood-Schätzer bzw. -Modelle ermöglichen es, z.B ein multiplikatives 
Modell mit konstanten Variationskoeffizienten ( 2und0 == ψξ ) zu modellieren, ohne sich 
über Verteilungsannahmen Gedanken zu machen. 
 
Für die technische Umsetzung ist es hilfreich, dass für Power-Varianz-Funktionen die 
Devianz (3) direkt in Abhängigkeit von ψ  ausgedrückt werden kann, wobei die Link-Funktion 

über die Erwartungswertbeziehung aus (1) implizit in iµ  enthalten ist: 

Für die beiden Ausnahmen ψ  = 1,2 sei auf die Devianz-Formeln in Tab. 2 verwiesen. Die 
Sensitivität von Modellen gegenüber inkrementellen Änderungen in der Varianz- und Link-
Funktion kann analytisch und graphisch über die Veränderungen der skalierten Quasi-
Devianz ),,;(* ψξµ yDQ  mit 

beurteilt werden. Zu dem ersten Term, der skalierten Devianz aus (4a), wird ein von der 
Modellschätzung unabhängiger Korrekturfaktor addiert, der erst den Vergleich über 
verschiedene Varianzfunktionen ermöglicht. 
 

 
Abb. 1a) Devianz-Profile für zwei Link-Fkt. 

 Abb. 1b) Dreidimensionales Devianz-Profil 
 

Die Veränderung der skalierten Quasi-Devianz ),,;(* ψξµ yDQ  in Abhängigkeit von ψξ bzw.  

lässt sich anschaulich in Devianz-Profil-Plots abbilden. Abb. 1a zeigt ein zwei-dimensionales 
Devianz-Profil für zwei gegebene ξ bei variierenden ψ . Abb. 1b zeigt ein 
dreidimensionales Devianz-Profil bei der simultanen Variation von ψξ und . 
 
Modellierung von GLMs mit PROC GENMOD 
 
Von der Struktur gleicht PROC GENMOD anderen SAS–Prozeduren für lineare Modelle, wie 
z.B. Proc GLM oder Proc REG. Zwei interessante Optionen von PROC GENMOD sind die 
Modellierung von benutzerdefinierten Link- und Varianzfunktionen und die Performance-
Optimierung durch Vorgabe von Startwerten für die Parameterberechnung. 
 
Ausgehend von dem Gamma-Modell mit logarithmischer Linkfunktion wird in mehreren 
Schritten gezeigt, wie Power-Link- bzw. Power-Varianz-Funktionen in PROC GENMOD 
implementiert werden können, um die Datenpunkte für Devianz-Profil-Plots zu berechnen.  
Bei der Modellierung mit PROC GENMOD (Beispiel 1) müssen neben der z.B. aus PROC 
GLM bekannten MODEL-Anweisung mit Angabe der Modellvariablen lediglich die Verteilung 
der Responsevariablen und die Linkfunktion (DIST- und LINK-Option in der MODEL-
Anweisung) definiert werden. Mit der SCWGT-Anweisung kann die Gewichtungsvariable w 
berücksichtigt werden. Über die MAKE- Anweisung werden alle wichtigen Ergebnisse, wie 
z.B. die geschätzten Parameter, zur weiteren Bearbeitung in SAS-Dateien abgelegt. 

/*********************************************/ 
/**      Bsp. 1 Gamma-Modell „Einfach“      **/ 
/*********************************************/ 
 
  proc genmod data=MYDATA; 
    class X1 X2 X3; 
    model Y = X1 X2 X3 / dist=gamma link=log; 
    scwgt W; 
  
   make 'modelfit' out = MODDAT; 
   make 'parameterestimates' out = PAREST; 
   make 'obstats' out = OBSDAT; 
  run; 
 
/*********************************************/ 

 Beispiel 1: Gamma-Modell „Einfach“ 

 
Durch die DIST-Option wird die für die Schätzung wichtige Devianz- bzw. Varianz-Funktion 
festgelegt. Im Beispiel 2 werden beide Funktionen für das Gamma-Modell aus Beispiel 1 
gemäß den Funktionen aus Tab. 1 und 2 direkt implementiert. Hierzu stehen in PROC 
GENMOD u.a. die Systemvariablen _RESP_ und _MEAN_ für die Responsevariable y bzw. 
ihren geschätzten Erwartungswert µ  zur Verfügung. 

 
/*********************************************/ 
/**    Bsp. 2 Gamma-Modell „Kompliziert“    **/ 
/*********************************************/ 
 
  proc genmod data=MYDATA; 
    class X1 X2 X3; 
 
    variance var = _MEAN_**2; 
    deviance dev = 2*((_RESP_-_MEAN_)/_MEAN_-log(_RESP_/_MEAN_)); 
 
    model Y = X1 X2 X3 / link=power(0); 
    scwgt W; 
  run; 
 
/*********************************************/ 

 Beispiel 2: Gamma-Modell „Kompliziert“ 

 
In Beispiel 3 werden die in (6) vorgestellten Power-Link- bzw. Power-Varianz-Funktionen 
implementiert. Power-Link-Funktionen werden über die LINK-Option POWER(Wert) 
abgebildet. Die entsprechende Devianz-Funktion aus (8) wird analog dem Beispiel 2 
implementiert. Zur besseren Strukturierung können alle relevanten Variablen und Formeln 
vorab definiert werden, um sie dann komprimiert in dem VARIANCE- und DEVIANCE-
Statement zusammenfließen zu lassen. So können z.B. IF-THEN-ELSE-Beziehungen für 
komplexere Devianz-Strukturen implementiert werden. In Beispiel 3 wird wiederum die 
Gamma-Verteilung mit logarithmischer Linkfunktion, d.h. 2 als Exponent der Power-Varianz-
Funktion und 0 als Exponent der Power-Link-Funktion, modelliert. 

/**********************************************/ 
/**  Bsp. 3 Gamma-Modell „sehr Kompliziert“  **/ 
/**********************************************/ 
 
  proc genmod data=MYDATA; 
    class X1 X2 X3; 
      P = 2; 
      M = _MEAN_; 
      Y = _RESP_; 
      If _Y = 0 then _D = 0; 
      else if p=1 then D=2*(_Y*log(Y/A)-(Y-A)); 
      else if P=2 then D=2*( -log(Y/A) + (Y - A)/A); 
      else  D = 2 * ( (-Y)/(1-p)*(a**(1-p)-y**(1-p))  
                 + 1/(2-p)*(a**(2-p)-y**(2-p))); 
    variance var = A**2; 
    deviance dev = D; 
    model Y = X1 X2 X3 / link=power(0) ; 
    scwgt W; 
  run; 
 
/**********************************************/ 

 Beispiel 3: Gamma-Modell „sehr kompliziert“ 

 
Zur Datengenerierung für Devianz-Profil-Plots muss eine Vielzahl von Modellen iterativ 
berechnet werden. Über ein Macro kann diese Aufgabe automatisiert werden, wobei die 
Indizes für die Power-Link- und Power-Varianz-Funktionen als Laufindizes der Iterationen 
dienen. Die Berechnung der skalierten Quasi-Devianz-Werte ),,;(* ψξµ yDQ  gemäß den 

Formeln (5b), (8) und (9) erfolgt durch Auslesen der Modellinformationen über die MAKE-
Anweisung und Weiterverarbeitung in einem DATA-Step. Die graphische Darstellung der 
Devianz-Profile (vgl. Abb. 1a,b) kann z.B. über SAS-GRAPH erfolgen. 
 

 
   Abb. 2a) Devianz-Profil ohne Startwerte 

Abb. 2b) Devianz-Profil mit Startwerte 
 
Je weiter ψξ und  von ihren Devianz-minimierenden Optima entfernt liegen, desto größer ist 
die Wahrscheinlichkeit, dass die Modelle nicht bzw. nur an einem lokalen Optimum konver-
gieren. Die hier ermittelten Quasi-Devianz-Werte ),,;(* ψξµ yDQ  werden als extreme (nicht 

verwertbare) Ausreißer in den Plots (vgl. Abb. 2a) sichtbar. Diese Probleme werden häufig 
von folgenden Warn- oder Fehlermeldungen im SAS-Log-File begleitet: 

Beispiel 4: Fehlermeldungen in PROC GENMOD 

 
Leider stellt PROC GENMOD keine Systemvariable o.ä. zur Verfügung, um dieses Problem 
anzuzeigen. Diese Probleme in der Modellschätzung können in der MODEL-Anweisung mit 
der Option STRTVALS zur Berücksichtigung von Startwerten begrenzt werden. Ziel ist es, 
Startwerte zu finden, die sich bereits in Nähe der endgültigen Parameterschätzer befinden. 
Es liegt auf der Hand, dass sich die Modellparameter bei einer inkrementellen Änderung des 
Index ψ  der Power-Varianz-Funktion i.d.R. nur marginal ändern. Die Parameter einer 
Iteration sind also sinnvolle Startwerte der jeweils nächsten Iteration. Beispiel 5 zeigt ein 
einfaches iteratives Makro zur effizienten Nutzung dieser Option STRTVALS.  
 

/********************************************************/ 
/**  Bsp. 5 Verwendung von Startwerten in einem Makro  **/ 
/********************************************************/ 
 
%macro QUASI; 
 
%let startwerte=; /*Erste Iteration hat keine Startwerte*/ 
%do index = -1 %to 1 %by 1;  /*Laufindex für Power-Link */ 
 
  proc genmod data=MYDATA; 
      [Weiterer Code wie Bsp. 3] 
   model Y = X1 X2 X3 / link = pow(&index)  &startwerte; 
   make 'parameterestimates' out = PAREST; /*S.-W. nächste It.*/ 
  run; 
 
/***  Auslesen der Modellparameter in Macro-Variablen ***/; 
   data  _null_;  
   set   PAREST; 
      call symput( 'B'||left(_N_), put( estimate, 16.8 ) ); 
      call symput( 'parms', put( _N_, 4.) ); 
   run; 
 
/****     Text-String der Startwerte generieren     *****/ 
     %let parms = %eval(&parms-1);   /*Letzter Parm 'weg*/ 
     %let startwerte = intercept=&B1  initial=  ; 
        %do i=2 %to &parms; 
           %let startwerte=&startwerte &&B&i; 
        %end; 
    run; 
 %end;          /* Ende der Schleife */; 
%mend QUASI;      /* Ende des Macros */ 
 
/********************************************************/ 

 Beispiel 5: Verwendung von Startwerten 

 

In dem Makro QUASI_LH im Anhang wird Beispiel 5 um die Schleife für die Power-Varianz-
Funktion und die DATA-Steps zur Berechnung und Ausgabe der Quasi-Devianz-Werte 

),,;(* ψξµ yDQ  erweitert. Es empfiehlt sich, die Schleife für die Power-Varianz-Funktion als 

innere Schleife zu programmieren. Die Laufindizes sollten so gewählt werden, dass die Itera-

tionen sich von einer sicher konvergierenden Ausgangskombination der Indizes ψξ bzw.  in 
die Randbereiche vorarbeiten.  
 
Ein positiver Nebeneffekt der Verwendung von Startwerten ist die teilweise erhebliche 
Reduktion der Rechenzeit. In Abhängigkeit der Datenkonstellation und der Güte der Start-
werte ist mit einer Verringerung der CPU-Zeit zwischen 20 und 50 Prozent zu rechnen. Die 
Verwendung von Startwerten kann z.B. auch bei der schrittweisen Variablenauswahl, wie sie 
u.a. in PROC REG fest implementiert ist, sinnvoll eingesetzt werden. 

Schlussbemerkung 
Im Bereich der linearen Modelle stellt PROC GENMOD ein sehr vielseitiges und leistungs-
fähiges Werkzeug dar. Die Stärke von PROC GENMOD liegt weniger in der Tiefe der Funk-
tionalität für einzelne Teilfragen. Diese werden durch die auf das jeweilige lineare Modell 
zugeschnittenen Prozeduren, wie z.B. PROC REG oder PROC PROBIT besser abgedeckt. 
Die Stärken liegen vor allem im explorativen Bereich bei Modellbewertung und -auswahl.  

Anhang: SAS-Macro QUASI_LH 

%macro QUASI_LH(InData=,OutFile=, Y_Var=, X_Vars=, Ex_Var=,Cls_Vars=, 
                  StartV=, StopV= ,IncrV=, StartL=, StopL= ,IncrL= ); 
/*******************************************************************/ 
/*           AUTHOR : Olaf Kruse, Summer 1995                      */ 
/*                    Revised Spring 2001                          */ 
/*******************************************************************/ 
/*  USAGE:                                                         */ 
/*   %QUASI_LH(InData   = <Input Data>,                            */ 
/*             OutData  = <Output File>                            */ 
/*             Y_Var    = <Response-Variable>,                     */ 
/*             X_Vars   = <Prediktoren>,                           */ 
/*             Cls_Vars = <Class-Variablen, Faktoren >,            */ 
/*             Ex_Var   = <Beobachtungs-Gewichte >,                */ 
/*   StartV = <Startwert*100 der Power-Varianz-Funktion>           */ 
/*   StopV  = <Endwert*100 der Power-Varianz-Funktion>             */ 
/*   IncrV  = <Iterationsschrittweite*100 für Power-Varianz>       */ 
/*   StartL = <Startwert*100 der Power-Link-Funktion>              */ 
/*   StopL  = <Endwert*100 der Power-Link-Funktion>                */ 
/*   IncrL  = <Iterationsschrittweite*100 für Power-Link>          */ 
/*******************************************************************/ 
/*****************    Allgemeine Vorbereitungen    *****************/ 
ods listing close; 
%if %upcase(&Cls_Vars) ne %then %let clsstmt = %str(CLASS &Cls_Vars;); 
%else %let clsstmt= ; 
%let StrtVals= ; 
%let IncrV = %sysfunc(Abs(&IncrV)); 
%let IncrL = %sysfunc(Abs(&IncrL)); 
%if &StartL > &StopL %then %let IncrL = %eval(&IncrL*(-1)); 
%if &StartV > &StopV %then %let IncrV = %eval(&IncrV*(-1)); 
 
DATA &OutFile; 
set _null_; 
run; 
/*********   Äußere Schleife: Power-Link-Funktion          *********/ 
/*********   Innere Schleife: Power-Varianz-Funktion       *********/ 
/*********                  Hinweis:                  -    *********/ 
/*********   Schleifen-Reihenfolge kann durch Tauschen der *********/ 
/*********   nächsten beiden Code-Zeilen geändert werden!  *********/ 
 
%do vau = &StartL %to &StopL %by &IncrL;         
   %do tau = &StartV %to &StopV %by &IncrV;      
 
   %let beta = %sysevalf(&vau/100);/* Laufindex Power-Link-Fkt.    */ 
   %let alpha =%sysevalf(&tau/100);/* Laufindex Power-Varianz-Fkt. */ 
   %put **Log-File:  Link: &beta  Var: &alpha  **;  
 
   proc genmod data=&InData; 
      &clsstmt 
      make 'modelfit' out = MODDAT; 
      make 'parameterestimates' out = PAREST; 
      make 'obstats' out = OBSDAT; 
      _P = &alpha; 
      _A = _MEAN_; 
      _Y = _RESP_; 
      If _Y = 0 then _D = 0; 
         else if _P=1 then _D=2*(_Y*log(_Y/_A)-(_Y-_A)); 
         else if _P=2 then _D=2*((_Y-_A)/_A-log(_Y/_A)); 
         else _D = 2 * ( (-_Y)/(1-_P)*(_A**(1-_P)-_Y**(1-_P)) 
                         + 1/(2-_P)*(_A**(2-_P)-_Y**(2-_P))); 
      variance _VAR = _A**_P; 
      deviance _DEV = _D; 
      scwgt  &Ex_Var; 
      model &Y_Var = &X_Vars / link = pow(&beta) obstats &StrtVals ; 
      run; 

/*********        String der Startwerte generieren         *********/ 
      data  _null_ ; 
      set   PAREST; 
         call symput( ’B’||left(_N_), put( estimate, 16.8 ) ); 
         call symput( ’parms’, put( _N_, 4.) ); 
         run; 
      %let parms = %eval(&parms-1); /*Letzter Parm ’weg*/ 
      %let Strtvals= intercept=&B1  initial=  ; 
         %do i=2 %to &parms; 
            %let Strtvals=&Strtvals &&B&i; 
         %end; 
       run; 
/******  Berechnen der Werte der Extended Quasi-Devianz  ***********/ 
/*******   Auslesen der Anzahl Freiheitsgrade für Teta   ***********/ 
      data _NULL_; 
      set MODDAT; 
         if _N_ = 1 then call symput( 'DF', put( df, 10. ) ); 
         run;      /***  Anzahl Freiheitsgrade für Teta  ***/ 
/******  Berechnen der Werte der Extended Quasi-Devianz  ***********/ 
      data ExQL; 
      merge &InData(keep= &Y_Var &Ex_Var  
                    rename=(&Y_Var=y &Ex_Var=wgt )) 
            obsdat(keep=pred rename=(pred=a)); 
         p=&alpha; 
         if wgt= 0 or y=0 then do; 
            nobs=0; /* Anzahl gültiger Beobachtungen */ 
            Chi2=0; /* Für Teta */ 
            Dev=0;  /* Normale Devianz */ 
         end; 
         else do; 
            if p=1 then Dev =2*wgt*(y*log(y/a)-(y-a)); 
            else if p=2 then Dev =2*wgt*((y-a)/a-log(y/a)); 
            else Dev =2*wgt*((-y)/(1-p)*(a**(1-p)-y**(1-p))+ 
                               1/(2-p)*(a**(2-p)-y**(2-p))); 
            Chi2=wgt*(y-a)**2/a**p; 
            nobs=1; 
            DY=log(6.2832/wgt*y**p)/* Für Korrekturterm */; 
         end; 
         run; 
 
   proc means sum noprint data=ExQL; 
      var Dev DY Chi2 nobs; 
      output out=statistics sum=; 
      run; 
 
   data statistics(keep=PLink PVar Teta D1 D2 QDev ); 
      retain PLink PVar Teta D1 D2 QDev ; 
   set statistics; 
      PLink=&beta;          /* Power-Link-Index */ 
      PVar =&alpha;         /* Power-Var-Index */ 
      Teta=Chi2/&DF;        /* Dispersions-Par. */ 
      D1=Dev/Teta;          /* skalierte Dev.   */ 
      D2=DY+nobs*log(teta); /* Korrekturterm    */ 
      QDev=D1+D2;           /* Quasi-Devianz    */ 
      run; 
 
   data &OutFile; /* Ergebnisse der Iterationen zusammenführen */ 
   set &OutFile statistics; 
   run; 
 
   %end;  * innere Schleife *; 
%end;  * äußere Schleife *; 
 
ods listing; 
 
%mend QUASI_LH; 


